

STA431 Handout 9

Double Measurement Regression on the BMI Data

```

/***** bmi5.sas *****/
options linesize=79 pagesize = 500 noovp formdlim='-';
title 'BMI and Health: Double measurement Regression';

data health;
  infile 'bmihealth.data';
  input age1 bmi1 fat1 cholest1 diastol1
        age2 bmi2 fat2 cholest2 diastol2;
        /* fat1 and fat2 are percent body fat */
  age = (age1+age2)/2; bmi = (bmi1+bmi2)/2; fat = (fat1+fat2)/2;
  cholest = (cholest1+cholest2)/2 ; diastol = (diastol1+diastol2)/2;

proc calis cov;
  /* Analyze the covariance matrix (Default is corr). */
  title2 'Full Model';
  var age1 -- diastol2; /* Name the observed variables */
  /* Now give simultaneous equations, separated by commas. Latent
  variables begin with F for factor. Error terms begin with
  E for error or D for disturbance. SAS is not case sensitive.
  You must name all the parameters. Optional starting values in
  parentheses may be given after the parameters. */
  lineqs
    Fcholest = gamma11 Fage + gamma12 Fbmi + gamma13 Ffat + e1,
    Fdiastol = gamma21 Fage + gamma22 Fbmi + gamma23 Ffat + e2,
    age1      = Fage      + delta11,
    bmi1      = Fbmi      + delta12,
    fat1      = Ffat      + delta13,
    cholest1  = Fcholest + eps11,
    diastol1  = Fdiastol + eps12,
    age2      = Fage      + delta21,
    bmi2      = Fbmi      + delta22,
    fat2      = Ffat      + delta23,
    cholest2  = Fcholest + eps21,
    diastol2  = Fdiastol + eps22;
  std /* Variances (not standard deviations). Colon means fill
        in the numbers. */
    Fage Fbmi Ffat = 3 * phi: ,
    e1 e2 = 2 * psi: ,
    delta11 delta12 delta13 eps11 eps12 = 5 * omega1_: ,
    delta21 delta22 delta23 eps21 eps22 = 5 * omega2_: ;
  cov /* Covariances: If not mentioned, it's zero. */
    Fage Ffat Fbmi = 3 * phi_ij: , e1 e2 = psi12 ,
    delta11 delta12 delta13 eps11 eps12 = 10 * omega1_ij: ,
    delta21 delta22 delta23 eps21 eps22 = 10 * omega2_ij: ;
  bounds 0.0 < phi1-phi3 psi1 psi2 omega1_1-omega1_5 omega2_1-omega2_5;
        /* Variances are positive */

/* Now fit a reduced model to test H0: gamma12 = gamma22 = 0,
meaning BMI is unrelated to either cholesterol or blood pressure
if we allow for age and percent body fat. Use psummary to see
just the minimal output needed for a likelihood ratio test. */
```

```

proc calis cov psummary;
  title2 'Reduced Model for testing H0: gamma12 = gamma22 = 0';
  var age1 -- diastol2;
  lineqs
    Fcholest = gamma11 Fage + gamma12 Fbmi + gamma13 Ffat + e1,
    Fdiastol = gamma21 Fage + gamma22 Fbmi + gamma23 Ffat + e2,
    age1      = Fage      + delta11,
    bmi1      = Fbmi      + delta12,
    fat1      = Ffat      + delta13,
    cholest1  = Fcholest + eps11,
    diastol1  = Fdiastol + eps12,
    age2      = Fage      + delta21,
    bmi2      = Fbmi      + delta22,
    fat2      = Ffat      + delta23,
    cholest2  = Fcholest + eps21,
    diastol2  = Fdiastol + eps22;
  std
    Fage Fbmi Ffat = 3 * phi: ,
    e1 e2 = 2 * psi: ,
    delta11 delta12 delta13 eps11 eps12 = 5 * omega1_: ,
    delta21 delta22 delta23 eps21 eps22 = 5 * omega2_: ;
  cov
    Fage Ffat Fbmi = 3 * phi_ij: , e1 e2 = psi12 ,
    delta11 delta12 delta13 eps11 eps12 = 10 * omega1_ij: ,
    delta21 delta22 delta23 eps21 eps22 = 10 * omega2_ij: ;
  bounds 0.0 < phi1-phi3 psi1 psi2 omega1_1-omega1_5 omega2_1-omega2_5;
  lincon gamma12=0, gamma22=0; /* Constrain model to obey this H0 */

  /* Using lincon is much safer than fitting a model with the variables
  just missing in the regression equations. However, SAS will warn
  us that "There are 2 active constraints at the solution," and carry
  on in a pretty menacing way. The warning can be ignored, in this
  particular case where the linear constraints are setting
  parameters exactly equal to zero. Do NOT ignore the warning
  if an inequality is involved. */

proc iml;
  title2 'Calculate Likelihood ratio test of H0: gamma12=gamma22=0';
  G = 19.5309 - 18.9011;
  /* Difference between chisquares (copied from the printout) */
  pval = 1 - probchi(G,2);
  print G,pval;
  print " ";

/* You could also do this with R or a calculator. */

```

Log file says

WARNING: There are 2 active constraints at the solution. The standard errors and Chi-Square test statistic assume the solution is located in the interior of the parameter space and hence do not apply if it is likely that some different set of inequality constraints could be active.

NOTE: The degrees of freedom are increased by the number of active constraints (see Dijkstra, 1992). The number of parameters in calculating fit indices is decreased by the number of active constraints. To turn off the adjustment, use the NOADJDF option.

Some parts of the list file are skipped.

BMI and Health: Double measurement Regression 1
Full Model

The 12 Endogenous Variables

Manifest	age1	bmi1	fat1	cholest1	diastol1	age2
	bmi2	fat2	cholest2	diastol2		
Latent	Fcholest	Fdiastol				

The 15 Exogenous Variables

Manifest						
Latent	Fage	Fbmi	Ffat			
Error	e1	e2	eps11	eps12	eps21	
	eps22	delta11	delta12	delta13	delta21	
	delta22	delta23				

BMI and Health: Double measurement Regression 4
Full Model

The CALIS Procedure

Covariance Structure Analysis: Maximum Likelihood Estimation

Observations	500	Model Terms	1
Variables	10	Model Matrices	4
Informations	55	Parameters	45

Variable	Mean	Std Dev
age1	43.08800	13.01257
bmi1	25.43580	4.55383
fat1	18.79000	7.76001
cholest1	262.02740	52.55520
diastol1	88.42400	19.29136
age2	44.40600	12.43005
bmi2	25.51760	3.72919
fat2	18.90280	7.57154
cholest2	261.24560	53.92955
diastol2	88.70000	13.09921

NOTE: Some initial estimates computed by two-stage LS method.

BMI and Health: Double measurement Regression
Full Model

6

The CALIS Procedure
Covariance Structure Analysis: Maximum Likelihood Estimation

Dual Quasi-Newton Optimization

Dual Broyden - Fletcher - Goldfarb - Shanno Update (DBFGS)

Parameter Estimates 45
Functions (Observations) 55
Lower Bounds 15
Upper Bounds 0

Optimization Start

Active Constraints 0 Objective Function 0.3271903962
Max Abs Gradient Element 0.0797311211

Iter	Rest arts	Func Calls	Act Con	Objective Function	Obj Fun Change	Max Abs Gradient Element	Step Size	Slope Search Direc
1	0	3	0	0.30497	0.0222	0.2449	0.0466	-1.030
2	0	4	0	0.28490	0.0201	0.1134	0.320	-0.173
3	0	5	0	0.26879	0.0161	0.0925	1.000	-0.0413

skipping ...

165	0	291	0	0.03788	0.000016	0.000727	72.118	-457E-9
166	0	293	0	0.03788	1.362E-6	0.000013	1.026	-265E-8
167	0	295	0	0.03788	4.984E-8	0.000036	3.287	-3E-8
168	0	296	0	0.03788	4.69E-8	0.000015	3.018	-41E-9
169	0	298	0	0.03788	4.252E-9	9.062E-6	1.135	-75E-10

Optimization Results

Iterations 169 Function Calls 299
Gradient Calls 240 Active Constraints 0
Objective Function 0.0378778957 Max Abs Gradient Element 9.0624024E-6
Slope of Search Direction -7.494794E-9

ABSGCONV convergence criterion satisfied.

BMI and Health: Double measurement Regression
Full Model

7

The CALIS Procedure
Covariance Structure Analysis: Maximum Likelihood Estimation

Fit Function	0.0379
Goodness of Fit Index (GFI)	0.9926
GFI Adjusted for Degrees of Freedom (AGFI)	0.9591
Root Mean Square Residual (RMR)	19.5113
Parsimonious GFI (Mulaik, 1989)	0.2206
Chi-Square	18.9011
Chi-Square DF	10
Pr > Chi-Square	0.0415
Independence Model Chi-Square	4015.9
Independence Model Chi-Square DF	45
RMSEA Estimate	0.0422
RMSEA 90% Lower Confidence Limit	0.0081
RMSEA 90% Upper Confidence Limit	0.0711
ECVI Estimate	0.2223
ECVI 90% Lower Confidence Limit	0.2051
ECVI 90% Upper Confidence Limit	0.2554
Probability of Close Fit	0.6316
Bentler's Comparative Fit Index	0.9978
Normal Theory Reweighted LS Chi-Square	18.6927
Akaike's Information Criterion	-1.0989
Bozdogan's (1987) CAIC	-53.2450
Schwarz's Bayesian Criterion	-43.2450
McDonald's (1989) Centrality	0.9911
Bentler & Bonett's (1980) Non-normed Index	0.9899
Bentler & Bonett's (1980) NFI	0.9953
James, Mulaik, & Brett (1982) Parsimonious NFI	0.2212
Z-Test of Wilson & Hilferty (1931)	1.7350
Bollen (1986) Normed Index Rho1	0.9788
Bollen (1988) Non-normed Index Delta2	0.9978
Hoelter's (1983) Critical N	485

BMI and Health: Double measurement Regression
Full Model

8

The CALIS Procedure
Covariance Structure Analysis: Maximum Likelihood Estimation

Manifest Variable Equations with Estimates

age1	=	1.0000	Fage	+	1.0000	delta11
bmi1	=	1.0000	Fbmi	+	1.0000	delta12
fat1	=	1.0000	Ffat	+	1.0000	delta13
cholest1	=	1.0000	Fcholest	+	1.0000	eps11
diastol1	=	1.0000	Fdiastol	+	1.0000	eps12
age2	=	1.0000	Fage	+	1.0000	delta21
bmi2	=	1.0000	Fbmi	+	1.0000	delta22
fat2	=	1.0000	Ffat	+	1.0000	delta23
cholest2	=	1.0000	Fcholest	+	1.0000	eps21
diastol2	=	1.0000	Fdiastol	+	1.0000	eps22

BMI and Health: Double measurement Regression
Full Model

9

The CALIS Procedure
Covariance Structure Analysis: Maximum Likelihood Estimation

Latent Variable Equations with Estimates

Fcholest =	0.1480*	Fage	+	-1.2703*	Fbmi	+	2.2271*	Ffat
Std Err	0.2128	gamma11		1.6550	gamma12		0.8974	gamma13
t Value	0.6957			-0.7675			2.4818	
			+	1.0000	e1			

Fdiastol =	0.0104*	Fage	+	-0.1054*	Fbmi	+	1.2692*	Ffat
Std Err	0.0458	gamma21		0.3941	gamma22		0.2044	gamma23
t Value	0.2269			-0.2674			6.2090	
			+	1.0000	e2			

VariANCES of Exogenous Variables

Variable	Parameter	Estimate	Standard Error	t Value
Fage	phi1	148.42528	9.81374	15.12
Fbmi	phi2	12.56690	0.94238	13.34
Ffat	phi3	47.09143	3.29507	14.29
e1	psi1	2396	161.85535	14.81
e2	psi2	75.84924	8.80013	8.62
eps11	omega1_4	196.19717	54.14132	3.62
eps12	omega1_5	208.12976	15.27861	13.62
eps21	omega2_4	336.70368	56.85898	5.92
eps22	omega2_5	25.25689	7.68825	3.29
delta11	omega1_1	22.35003	3.22668	6.93
delta12	omega1_2	8.62777	0.71848	12.01
delta13	omega1_3	13.98305	1.72274	8.12
delta21	omega2_1	6.31681	2.90687	2.17
delta22	omega2_2	1.28742	0.48644	2.65
delta23	omega2_3	11.48662	1.67843	6.84

Covariances Among Exogenous Variables

Var1	Var2	Parameter	Estimate	Standard Error	t Value
Fage	Fbmi	phi_ij2	8.55218	2.14675	3.98
Fage	Ffat	phi_ij1	28.72995	4.21197	6.82
Fbmi	Ffat	phi_ij3	21.11231	1.59012	13.28
e1	e2	psi12	27.04719	23.18142	1.17
eps11	eps12	omega1_ij10	2.37831	17.47881	0.14
eps21	eps22	omega2_ij10	10.57885	12.16913	0.87
eps11	delta11	omega1_ij4	1.46892	9.30486	0.16
eps12	delta11	omega1_ij7	5.51726	4.26881	1.29
eps11	delta12	omega1_ij5	-1.69104	4.06349	-0.42
eps12	delta12	omega1_ij8	8.24458	2.39020	3.45
delta11	delta12	omega1_ij1	2.41623	1.00309	2.41
eps11	delta13	omega1_ij6	-1.68245	6.64325	-0.25
eps12	delta13	omega1_ij9	0.90754	3.58593	0.25
delta11	delta13	omega1_ij2	-1.23971	1.62508	-0.76
delta12	delta13	omega1_ij3	8.09346	0.97038	8.34
eps21	delta21	omega2_ij4	0.83367	9.04174	0.09
eps22	delta21	omega2_ij7	2.25362	2.79309	0.81
eps21	delta22	omega2_ij5	0.09766	3.35074	0.03
eps22	delta22	omega2_ij8	2.97847	1.44756	2.06
delta21	delta22	omega2_ij1	0.26709	0.76879	0.35
eps21	delta23	omega2_ij6	-4.04534	6.67275	-0.61
eps22	delta23	omega2_ij9	2.33950	2.56766	0.91
delta21	delta23	omega2_ij2	1.97628	1.52225	1.30
delta22	delta23	omega2_ij3	0.06910	0.74472	0.09

BMI and Health: Double measurement Regression 12
Reduced Model for testing H0: $\gamma_{12} = \gamma_{22} = 0$

The CALIS Procedure
Covariance Structure Analysis: Maximum Likelihood Estimation

WARNING: There are 2 active constraints at the solution. The standard errors and Chi-Square test statistic assume the solution is located in the interior of the parameter space and hence do not apply if it is likely that some different set of inequality constraints could be active.

NOTE: The degrees of freedom are increased by the number of active constraints (see Dijkstra, 1992). The number of parameters in calculating fit indices is decreased by the number of active constraints. To turn off the adjustment, use the NOADJDF option.

BMI and Health: Double measurement Regression 13
Reduced Model for testing H0: $\gamma_{12} = \gamma_{22} = 0$

The CALIS Procedure
Covariance Structure Analysis: Maximum Likelihood Estimation

Fit Function	0.0391
Goodness of Fit Index (GFI)	0.9924
GFI Adjusted for Degrees of Freedom (AGFI)	0.9650
Root Mean Square Residual (RMR)	19.8115
Parsimonious GFI (Mulaik, 1989)	0.2646
Chi-Square	19.5309
Chi-Square DF	12
Pr > Chi-Square	0.0765
Independence Model Chi-Square	4015.9
Independence Model Chi-Square DF	45

skipping ...

BMI and Health: Double measurement Regression 14
Calculate Likelihood ratio test of H0: $\gamma_{12}=\gamma_{22}=0$

G
0.6298
PVAL
0.7298619